

**Pselnt (FOR - para)**

Has de lliurar el codi de cada algorisme en **format psc** degudament comentat. Per fer aquesta tasca pots modificar els exercicis de la tasca anterior i desar-los amb un altre nom.

1. Fes un algorisme que demana 10 números per teclat i mostri per pantalla, per cada número entrat, si és parell o és senar.

```
1  Algoritmo pr4_ex1
2  Definir i, num Como Entero
3  Definir MAX_NUM Como Entero
4  MAX_NUM = 10
5
6  Para i←0 Hasta MAX_NUM-1 Hacer
7  |   Escribir "Introdueix un número: "
8  |   Leer num
9  |
10 |   Si num MOD 2 == 0
11 |   |   Escribir "El número " num " és parell."
12 |   SiNo
13 |   |   Escribir "El número " num " és imparell."
14 |   FinSi
15 FinPara
16 FinAlgoritmo
```

2. Fes l'exercici de mostrar totes les taules de multiplicar de l'1 al 10 amb un bucle para aniuat.

```
1  Algoritmo pr4_ex2
2  Definir i, j Como Entero
3  Definir MAX_TAULA Como Entero
4  MAX_TAULA = 10
5
6  Para i = 1 Hasta MAX_TAULA Hacer
7  |   Escribir "Taula del " i
8  |   Para j = 1 Hasta MAX_TAULA Hacer
9  |   |   Escribir i "x" j "=" i*j
10 |   FinPara
11 |   Escribir ""
12 FinPara
13
14 FinAlgoritmo
```

3. Fes un programa que simuli el llançament de 2 daus 100 vegades. Calcula la mitjana de tots els llançaments.

```
1  Algoritmo pr4_ex3|
2  Definir NUM_TIRADES, i, dau_1, dau_2 Como Entero
3  Definir mitjana Como Real
4
5  NUM_TIRADES = 100
6  mitjana = 0
7
8  Escribir "Tirant dos daus 100 vegades..."
9  Para i<-0 Hasta NUM_TIRADES-1 Hacer
10     dau_1 = Aleatorio(1, 6)
11     dau_2 = Aleatorio(1, 6)
12
13     mitjana = mitjana + dau_1 + dau_2
14  Fin Para
15
16  mitjana = mitjana / NUM_TIRADES
17  Escribir "La mitjana dels llencaments es " mitjana
18
19  FinAlgoritmo
20
```

4. Fes un algorisme que calculi tots els divisors d'un nombre.

```
1  Algoritmo pr4_ex4
2  Definir num, i Como Entero
3
4  Escribir "Introdueix un nombre: "
5  Leer num
6
7  Si num == 0 Entonces
8  |   Escribir "0 no te divisors"
9  SiNo
10 |   Escribir "Els divisors de " num " son: "
11 |   Para i=1 Hasta num Hacer
12 |   |   Si num MOD i == 0
13 |   |   |   //Divisor trobat
14 |   |   |   Escribir i " " Sin Saltar
15 |   |   FinSi
16 |   Fin Para
17 |   Escribir ""
18 |   FinSi
19 FinAlgoritmo
```


5. Fes un algorisme que comprovi si un número és primer o no.

```
1  Algoritmo pr4_ex5
2  Definir num, i Como Entero
3  Definir divisors Como Entero
4  divisors = 0
5
6  Escribir "Introdueix un nombre: "
7  Leer num
8
9  //Primer si només divisible entre ell mateix i la unitat
10 Si num <> 0 y num <> 1 Entonces
11     Para i←1 Hasta num Hacer
12         Si num MOD i == 0
13             //Divisor trobat
14             divisors = divisors + 1
15         FinSi
16     Fin Para
17 FinSi
18
19 Si num == 0 o num == 1 o divisors > 2 //No és primer
20     Escribir "El numero " num " no es primer."
21 SiNo
22     Escribir "El numero " num " es primer."
23 FinSi
24 FinAlgoritmo
```

6. Fes un algorisme que mostri per pantalla tots els primers entre 1 i 1000. Nota: L'1 no es considera primer per conveni.


```
1  Algoritmo pr4_ex6
2  Definir num, i Como Entero
3  Definir divisors, MAX_NUM Como Entero
4
5  MAX_NUM = 1000
6
7  Escribir "Mostrant els " MAX_NUM " primers nombres primers:"
8
9  Para num←1 Hasta MAX_NUM Hacer
10
11     //Primer si només divisible entre ell mateix i la unitat.
12     //L'1 no és primer per conveni.
13     divisors = 0
14     Para i←1 Hasta num Hacer
15         Si num MOD i == 0
16             //Divisor trobat
17             divisors = divisors + 1
18         FinSi
19     Fin Para
20
21     Si divisors == 2 //És primer
22         Escribir num " " Sin Saltar
23     FinSi
24
25 Fin Para
26 FinAlgoritmo
```

7. Fes el següent dibuix utilitzant un for



```
1  Algoritmo pr4_ex7
2      Definir MAX_COLS, MAX_FILS Como Entero
3      MAX_COLS = 10
4      MAX_FILS = 10
5      Definir SIMB1, SIMB2 Como Caracter
6      SIMB1 = "$"
7      SIMB2 = "-"
8      Definir i, j Como Entero
9
10     Para i = 1 Hasta MAX_FILS Con Paso 1 Hacer
11     |
12     |     Para j = 1 Hasta MAX_COLS Con Paso 2 Hacer
13     |     |     Si i MOD 2 == 0 Entonces
14     |     |     |     Escribir SIMB2 SIMB1 Sin Saltar
15     |     |     |     SiNo
16     |     |     |     Escribir SIMB1 SIMB2 Sin Saltar
17     |     |     FinSi
18     |     FinPara
19     |
20     |     Escribir ""
21     FinPara
22 FinAlgoritmo
```

8. fes el següent dibuix utilitzant un for





```
1  Algoritmo pr4_ex8
2  Definir MAX_COLS, MAX_FILS Como Entero
3  MAX_COLS = 10
4  MAX_FILS = 10
5  Definir SIMB1, SIMB2 Como Caracter
6  SIMB1 = " "
7  SIMB2 = "$"
8  Definir i, j, k Como Entero
9
10 Escribir "Versio 1:"
11 Para i = 1 Hasta MAX_FILS Hacer
12     //Escriure tants espais com número de 10-num fila
13     Para j = 1 Hasta MAX_COLS-i Con Paso 1 Hacer
14         Escribir SIMB1 Sin saltar
15     FinPara
16
17     //Escriure la diferència fins a 10 amb $
18     Para k = MAX_COLS-i+1 Hasta MAX_COLS Con Paso 1 Hacer
19         Escribir SIMB2 Sin Saltar
20     FinPara
21     Escribir ""
22 FinPara
23
24
25 Escribir ""
26 Escribir "Versio 2:"
27 Definir num_espais, num_dolars Como Entero
28
29 Para i = 1 Hasta MAX_FILS Hacer
30     num_dolars = i
31     num_espais = MAX_COLS - num_dolars
32
33     //Escriure tants espais com número de 10-num fila
34     Para j = 1 Hasta num_espais Con Paso 1 Hacer
35         Escribir SIMB1 Sin saltar
36     FinPara
37
38     //Escriure la diferència fins a 10 amb $
39     Para k = 1 Hasta num_dolars Con Paso 1 Hacer
40         Escribir SIMB2 Sin Saltar
41     FinPara
42     Escribir ""
43 FinPara
44 FinAlgoritmo
```



9. Fes el següent dibuix amb for



```

1  Algoritmo pr4_ex9
2  Definir POS_INI, MAX_FILES, MAX_VEGADES Como Entero
3  POS_INI = 10 //El primer $ s'escriu a la posició 10
4  MAX_FILES = 10
5  MAX_VEGADES = 10
6  Definir SIMB1, SIMB2 Como Caracter
7  SIMB1 = " "
8  SIMB2 = "$"
9  Definir i, j, k Como Entero
10
11  //1a línia: el 1r $ apareix a la posició central, la 10,
12  //2a línia: el 2n $ apareix a la posició central-1, la 9,
13  //etc...
14  //És a dir, a la 1a línia pintar 9 espais, 2a línia pintar 8,...
15  //Després de pintar els espais escriure "$ " tants cops
16  //com el número de línia actual
17  Para i = 1 Hasta MAX_FILES Hacer
18  |   //Escriure tants espais com 10-num fila
19  |   Para j = 1 Hasta POS_INI-i Con Paso 1 Hacer
20  |   |   Escribir SIMB1 Sin saltar
21  |   FinPara
22  |   //Escriure tants "$ " com el número de línia ens indiqui,
23  |   //com a màxim escriurem "$ " 10 cops, per tant
24  |   //fins a MAX_VEGADES
25  |   Para k = POS_INI-i+1 Hasta MAX_VEGADES Con Paso 1 Hacer
26  |   |   Escribir SIMB2 SIMB1 Sin Saltar
27  |   FinPara
28  |   Escribir ""
29  FinPara
30
31  FinAlgoritmo

```

10. Fes el següent dibuix amb for



```
1  Algoritmo pr4_ex10
2  Definir MAX_COLS, MAX_FILS Como Entero
3  MAX_COLS = 10
4  MAX_FILS = 10
5  Definir SIMB1, SIMB2 Como Caracter
6  SIMB1 = "*"
7  SIMB2 = " "
8  Definir i, j Como Entero
9
10 Para i = 1 Hasta MAX_FILS Hacer
11     Para j = 1 Hasta MAX_COLS Con Paso 1 Hacer
12         Si i == 1 o i == MAX_FILS Entonces
13             Escribir SIMB1 SIMB2 Sin saltar
14         SiNo
15             Si j == 1 o j == MAX_COLS Entonces
16                 Escribir SIMB1 SIMB2 Sin saltar
17             SiNo
18                 Escribir SIMB2 SIMB2 Sin saltar
19             FinSi
20         FinSi
21     FinPara
22     Escribir ""
23 FinPara
24
25 Escribir ""
26 FinPara
27
28 FinAlgoritmo
```