

Pselnt (WHILE)

Has de lliurar el codi de cada algorisme en **format psc** degudament comentat. Per fer aquesta tasca pots modificar els exercicis de la tasca anterior i desar-los amb un altre nom.

1. Fes un algorisme que vagi demanant números per teclat fins que introduïm un número negatiu i mostri per pantalla, per cada número entrat, si és parell o és senar.

Per l'entrada 5 mostrarà per consola:

```
* El número 5 és senar
```

```
1  Algoritmo pr2_ex1|
2      Definir num Como Entero
3
4      num = 0
5
6      Mientras num >= 0 Hacer
7          Escribir "Introdueix un nombre (negatiu per acabar): "
8          Leer num
9          Si num >= 0 Entonces
10             Escribir "El numero " num " es " Sin Saltar
11             Si num mod 2 == 0
12                 Escribir "parell"
13             SiNo
14                 Escribir "senar"
15             FinSi
16         FinSi
17     Fin Mientras
18
19 FinAlgoritmo
20
```

2. Fes un algorisme que imprimeix la taula de multiplicar dels nombres de l'1 al 10. Utilitza un while per mostrar la taula.

```
1  Algoritmo pr2_ex2
2  Definir i, j Como Entero
3  Definir MAX_TAULA Como Entero
4  MAX_TAULA = 10
5
6  i = 1
7
8  Mientras i <= MAX_TAULA Hacer
9      Escribir "Taula del " i
10     j = 1
11     Mientras j <= MAX_TAULA Hacer
12         Escribir i "x" j "=" i*j
13         j = j + 1
14     FinMientras
15     Escribir ""
16     i = i + 1
17 FinMientras
18
19 FinAlgoritmo
--
```

3. Fes un algorisme que calculi les n primeres potències d'un nombre. Haurem d'introduir per teclat el nombre i quantes potències volem calcular (màxim 5). Per exemple si li introduïm el 2 com a base, i el 5 com a potència, la consola mostrarà:

2 elevat a 0 és 1
2 elevat a 1 és 2
2 elevat a 2 és 4
2 elevat a 3 és 8
2 elevat a 4 és 16

```
1  Algoritmo pr2_ex3
2  |
3  Definir base, pot, res Como Entero
4  Definir i Como Entero
5
6  Escribir "Introdueix el nombre base: "
7  Leer base
8  Escribir "Introdueix la potencia (entre 1 i 5): "
9  Leer pot
10
11 Si pot ≤ 0 o pot > 5 Entonces
12     Escribir "La potencia ha de ser entre 1 i 5, ambdos inclosos"
13
14 SiNo
15     |
16     i = 0
17     Mientras i < pot Hacer
18         res = base^i
19         Escribir base " elevat a " i " es " res
20         i = i + 1
21     FinMientras
22 FinSi
23
24 FinAlgoritmo
```

4. Modifica l'algorisme d'endevinar un nombre. El programa demanarà nombres fins que l'encertis.

Fes un algorisme per endevinar un nombre entre l'1 i el 9 (ambdós inclosos) que es genera aleatòriament. L'usuari introduirà el nombre per teclat i el programa retornarà:

- si l'ha encertat: ENHORABONA!! Ets un crack!
- sinó. Si la diferència és només d'1:
 - quasi, pels pèls!
 - si la diferència és més 4: dedica't al parxís
 - si no: la propera vegada ho faràs millor

```
1  Algoritmo pr2_ex4
2  Definir num, num_rand, diff Como Entero
3  Definir trobat Como Logico
4
5  num_rand = Aleatorio(1, 9)
6  trobat = Falso
7
8  Mientras trobat == Falso Hacer
9
10     Escribir "Introdueix un numero entre 1 i 9: "
11     Leer num
12
13     diff = abs(num_rand - num)
14
15     Segun diff Hacer
16         0: Escribir "ENHORABONA!! Ets un crack!"
17             trobat = Verdadero
18
19         1: Escribir "quasi, pels pèls!"
20
21     De Otro Modo:
22         Si diff > 4 Entonces
23             Escribir "dedicat al parxís"
24         SiNo
25             Escribir "la propera vegada ho faras millor"
26         FinSi
27     Fin Segun
28
29 FinMientras
30
31 FinAlgoritmo
```

5. Fes un algorisme que comprovi la validesa d'un password, demanarem el password per teclat i hem de comprovar que:

- Comenci per la lletra 'A'
- Almenys tingui una xifra (del 0 al 9)
- Tingui una llargada mínima de 6 caràcters
- Tingui una llarga màxima de 16 caràcters
- Contingui 1 dels següents caràcters especials () / ! \$ % & "

L'usuari haurà de continuar introduint strings fins que n'introdueixi un de vàlid.

```
1  Algoritmo pr2_ex5
2
3  Definir pass, caract Como Caracter
4  Definir long, i, j Como Entero
5
6  Definir long_nums, long_espec Como Entero
7  Definir C_NUMS, C_ESPEC Como Caracter
8  C_NUMS = "0123456789"
9  C_ESPEC = "()/!$%&"
10
11 Definir conte_num, conte_espec Como Logico
12 conte_num = Falso
13 conte_espec = Falso
14
15 Escribir "Entra un password: "
16 Leer pass
17
18 long = Longitud(pass)
19 Si long < 6 o long > 16
20     Escribir "El password ha de tenir entre 6 i 16 caracters."
21
22 Sino
23
24     //Comprovar que comença per A abans del bucle
25     Si subcadena(pass, 0, 0) == 'A' Entonces
26
27         long_nums = Longitud(C_NUMS)
28         long_espec = Longitud(C_ESPEC)
29         i = 1
30         Mientras i < long Hacer
31             caract = subcadena(pass, i, i)
```

```
32
33      //Comprovar que tingui almenys una xifra
34      Si conte_num == Falso Entonces
35          j = 0
36          Mientras j < long_nums y conte_num == Falso Hacer
37              Si caract = subcadena(C_NUMS, j, j) Entonces
38                  // Ha trobar un número
39                  conte_num = Verdadero
40              SiNo
41                  j = j + 1
42              FinSi
43          FinMientras
44      FinSi
45
46      //Comprovar que tingui almenys un caracter especial
47      j = 0
48      Mientras j < long_espec y conte_espec == Falso Hacer
49          Si caract = subcadena(C_ESPEC, j, j) Entonces
50              // Ha trobar un caràcter especial
51              conte_espec = Verdadero
52          SiNo
53              j = j + 1
54          FinSi
55      FinMientras
56
57      i = i + 1
58  Fin Mientras
59
60      Si conte_num == Falso Entonces
61          Escribir "El password ha de contenir almenys una xifra."
62      FinSi
63
64      Si conte_espec == Falso Entonces
65          Escribir "El password ha de contenir almenys un caracter especial."
66      FinSi
67
68      Si conte_num == Verdadero y conte_espec == Verdadero
69          Escribir "El password es correcte."
70
71      FinSi
72
73      SiNo
74          Escribir "El password ha de començar per A."
75      FinSi
76
77  FinSi
78 FinAlgoritmo
```